

## Supplementary material S2 – example control stream

```
$PROBLEM EVEROLIMUS IN CANCER AND TRANSPLANT PATIENTS

$INPUT C ID TIME DV MDV EVID AMT CMT AGE LENG WT HT SEX OCC II SS PREDN

$DATA everolimus.csv IGNORE=C

$SUBROUTINE ADVAN5

$MODEL
COMP=(DOSE)
COMP=(LIVER)
COMP=(CENTRAL)
COMP=(PERIPH)
COMP=(TRAN)
COMP=(TRAN)
COMP=(TRAN)
COMP=(TRAN)

$PK

; --- TIME AFTER DOSE CALCULATION
IF (NEWIND.LE.1) THEN
DOSE=0
TDOS=0
ENDIF
IF (AMT.GT.0) THEN
DOSE=AMT
TDOS=TIME
ENDIF
TAD=TIME-TDOS

; --- FAT FREE MASS CALCULATION
IF (SEX.EQ.1) WHSMAX=42.92 ; MEN
IF (SEX.EQ.1) WHS50=30.93 ; MEN
IF (SEX.EQ.0) WHSMAX=37.99 ; WOMEN
IF (SEX.EQ.0) WHS50=35.98 ; WOMEN
HGT=LENG/100 ; HEIGHT CM->M
FFM=(WHSMAX*(HGT**2)*WT)/(WHS50*(HGT**2)+WT)

; --- ALLOMETRY SCALED TO 1.80M MAN OF 70 KG
ALLOCL=(FFM/57.18)**0.75
ALLOV=(FFM/57.18)

; --- LIVER VOLUME
BSA = ((WT**0.425)*(LENG**0.725))*0.007184 ; DUBOIS
POWER=1.108*(AGE**(-0.022))
VL=0.822*(BSA**POWER)

; --- FLAGS
FLA1=0
FLA2=0
IF (OCC.EQ.1) FLA1=1
IF (OCC.EQ.2) FLA2=1

; --- ABSORPTION
IOVMAT=(FLA1*ETA(3))+(FLA2*ETA(4))
F1=1
MAT=THETA(1)*EXP(IOVMAT)
KA=5/MAT

; --- DISTRIBUTION AND CLEARANCE OF EVEROLIMUS
FU=0.27*EXP(ETA(5))
QH=90*ALLOCL ; HEPATIC BLOOD FLOW L/H
QHP=QH*(1-HT) ; CALCULATE HEPATIC PLASMA FLOW
TVCLINT=THETA(2)*ALLOCL*(THETA(6)**PREDN) ; TYPICAL VALUE INTRINSIC CLEARANCE
CLINT=TVCLINT*EXP(ETA(1)) ; INTRINSIC CLEARANCE
TVV3=THETA(3)*ALLOV ; TYPICAL VALUE V3
V3=THETA(3)*EXP(ETA(2)) ; V3
```

```

Q=THETA(4)*ALLOCL ; INTERCOMPARTMENTAL CLEARANCE
TVV4=THETA(5)*ALLOV ; TYPICAL VALUE V4
V4=TVV4 ; V4
S3=V3 ; SCALING V3

; --- MASS TRANSPORT
EH=(CLINT*FU)/(QHP+(CLINT*FU)) ; HEPATIC EXTRACTION
CLH=EH*QHP ; HEPATIC CLEARANCE

K15=KA
K56=K15
K67=K56
K78=K67
K82=K78
K20=CLH/VL
K23=(QHP*(1-EH))/VL
K32=QHP/V3
K34=Q/V3
K43=Q/V4

$ERROR
CPL=F
BMAX=0.96425
KD=0.09195
KNS=0.15336
CRB=(BMAX*CPL)/(KD+CPL)+(KNS*CPL)
CBL=(CRB*HT)+(CPL*(1-HT))
RATIO=CRB/CPL
CPU=FU*CPL
IPRED=LOG(CBL+1E-16)
Y=IPRED+ERR(1) ; RESIDUAL ERROR MODEL

$THETA
(0.01, 0.404) ; MEAN ABSORPTION TIME THROUGH 4 TRANSIT COMPARTMENTS (H)
(1, 340) ; INTRINSIC CLEARANCE (L/H)
(1, 175) ; V3 CENTRAL PLASMA COMPARTMENT VOLUME (L)
(1, 85.7) ; INTERCOMPARTMENTAL CLEARANCE (L/H)
(1, 577) ; V4 PERIPHERAL PLASMA COMPARTMENT VOLUME (L)
(0.01, 1.31) ; HIGH DOSE PREDNISOLONE ON CLINT

$OMEGA
0.115 ; IIV CLINT
0.165 ; IIV V3
$OMEGA BLOCK(1) 1.21 ; IOV MAT
$OMEGA BLOCK(1) SAME ; IOV MAT
$OMEGA 0.0009 FIX ; IIV FU - KOVARIK ET AL

$SIGMA
0.032 ; PROP ERR
$EST METHOD=SAEM AUTO=1 PRINT=50
$EST METHOD=IMP PRINT=1 EONLY=1 NITER=5 ISAMPLE=1000
$COV MATRIX=R PRINT=E

```